

Better Titles:

Using The #BYVAR and #BYVAL Title Options

Arthur L. Carpenter
California Occidental Consultants

KEY WORDS

BY, #BYVAR, #BYVAL, BYLINE, TITLE

ABSTRACT

Although it has been possible to place macro variables in titles and footnotes for quite awhile, Releases 6.07 of the SAS® System and later include the ability to place BY variable values and names in the titles and footnotes as well. These options are only documented in SAS Technical Report P-222 and are therefore under utilized.

The names of BY variables and the BY variable values are inserted into titles and footnotes using the #BYVAR and #BYVAL options. There are two forms of these options and they will accommodate more than one BY variable. Several examples will be presented using these options.

INTRODUCTION

Prior to Release 6.07 of the SAS System, the use of a BY statement with a procedure that produced printed output always caused the BY-line to be written to the output. The BY-line consists of the variables (or for some procedures their labels) in the BY statement and the values taken on by those variables. Eliminating the BY-line and or moving the values of the BY variables into the titles has required the use of macro variables. Three new title options are

now available that can be used to make the process of controlling the BY-line and the BY variable values easier.

The BYLINE system option controls whether or not the BY-line will be displayed. A related title option #BYLINE places the BY-line in the title itself.

The name of variables in the BY statement can be added to a title by using the #BYVAR title option. The value taken on by the BY variable can also be added to a title through the use of the #BYVAL title option.

CONTROLLING THE BY-LINE

The following PROC PRINT uses a BY statement with one BY variable (STATION).

```
proc print
data=sasclass.h2oqual(obs=4);
by station;
var datetime depth temp ph;
title1 'Water Quality Data';
run;
```

This code produces the following output (using the NOCENTER option for this paper):

```
Water Quality Data
station=TS3
```

OBS	DATETIME	DEPTH	TEMP
1	06FEB93:09:15	0	13.6
2	06FEB93:09:15	1	13.5
3	06FEB93:09:15	2	13.5
4	06FEB93:09:15	3	14.1

Using the NOBYLINE system option will cause the BY-line to not be printed. This information could then be added to a title by using the #BYLINE title option. It is very important to notice that these title options all start with a pound sign (#).

```
options nobyline;

proc print
data=sasclass.h2oqual(obs=4);
by station;
var datetime depth temp;
title1 'Water Quality Data';
title2 'Information for #byline';
run;
```

This produces the following output:

```
Water Quality Data
Information for station=TS3
```

OBS	DATETIME	DEPTH	TEMP
1	06FEB93:09:15	0	13.6
2	06FEB93:09:15	1	13.5
3	06FEB93:09:15	2	13.5
4	06FEB93:09:15	3	14.1

INSERTING VARIABLE NAMES

The names of one or more of the variables in the BY statement can also be added into the title. The option is #BYVAR and it takes on two forms. You can name the variable explicitly or implicitly. In the following example titles 2 and 3 show these two forms. The #BYVAR option expects the NOBYLINE system option to be in effect for some procedures - see Technical Report P-222 for more details.

```
options nobyline;

proc print
data=sasclass.h2oqual(obs=4);
by station;
var datetime depth temp;
title1 'Water Quality Data';
title2 'Explicit - #byvar(station)';
title3 'Implicit - #byvar1';
run;
```

This results in the following output:

```
Water Quality Data
Explicit - station
Implicit - station
```

OBS	DATETIME	DEPTH	TEMP
1	06FEB93:09:15	0	13.6
2	06FEB93:09:15	1	13.5
3	06FEB93:09:15	2	13.5
4	06FEB93:09:15	3	14.1

At first it may seem rather silly to have an option of the form #BYVAR(station) in a title when what you want to show is 'station'. Why not just type in 'station' and be done with it? Although it is more likely that you will use the implicit form and just specify the position of the BY variable as was done in TITLE3 in the above example, the explicit form may be used when writing dynamic code. The option might become #BYVAR(&VAR), where the variable name to be stored in &VAR is unknown when the program is written.

INSERTING VARIABLE VALUES

When controlling the title text, it is very likely that you will want to insert the value of the BY variable for the current page. The #BYVAL option will do this for you. As with the #BYVAR option, it comes in two forms, explicit and implicit. In the following example the value of the first BY variable is requested implicitly (#BYVAL1). The same request could have been specified as #BYVAL(STATION).

```
proc print
data=sasclass.h2oqual(obs=4);
by station;
var datetime depth temp;
title1 'Water Quality Data';
title2 'Data Collected for #byvar1';
title3 '#byval1';
```

This produces the output:

Water Quality Data
Data Collected for station
TS3

OBS	DATETIME	DEPTH	TEMP
1	06FEB93:09:15	0	13.6
2	06FEB93:09:15	1	13.5
3	06FEB93:09:15	2	13.5
4	06FEB93:09:15	3	14.1

The option is smart enough to know to issue a page feed when the BY variable changes on a page with a title containing the #BYVAL option. For PROC PRINT this is the same as using the PAGEBY option.

SUMMARY

Adding the names and values of variables in a BY-line to titles became easier with Release 6.07 of the SAS® System. The #BYLINE, #BYVAR, and #BYVAL options can now be used to insert text and values directly into your titles.

REFERENCES

Additional examples and information on these options is available in the following publications by SAS Institute Inc.

Carpenter, Arthur L. and Charles E. Shipp, *Quick Results with SAS/GRAPH® Software*, Cary, NC: SAS Institute Inc., 1995. pp. 47-48, 87.

SAS Institute Inc., *SAS® Guide to Report Writing: Examples, Version 6, First Edition*, Cary, NC: SAS Institute Inc., 1994, pp. 29-34.

SAS Institute, Inc., *SAS Technical Report P-222, Changes and Enhancements to Base SAS Software, Release 6.07*, Cary, NC: SAS Institute Inc., 1991, pp. 44-45, 88, 159-162.

Westerlund, Earl, "Using BY Variable Text Substitution to Add Data Values to SAS/GRAPH Titles and Footnotes," in *Proceedings of the Nineteenth Annual SUGI Conference*, Cary, NC: SAS Institute Inc., 1994, pp. 153-154.

ABOUT THE AUTHOR



Art Carpenter's publications list includes two chapters in *Reporting from the Field*, the books *Quick Results with SAS/GRAPH® Software*, and *Carpenter's Complete Guide to the SAS® Macro Language*, and over two dozen papers and posters presented at SUGI, PharmaSUG, and WUSS. Art has been using SAS since 1976 and has served as a steering committee chairperson of both the Southern California SAS User's Group, SoCalSUG, and the San Diego SAS Users Group, SANDS; a conference cochair of the Western Users of SAS Software regional conference, WUSS; and Section Chair at the SAS User's Group International conference, SUGI.

Art is a SAS Quality Partner™ and through California Occidental Consultants he teaches SAS courses and provides contract SAS programming support nationwide.

Art Carpenter
California Occidental Consultants
P.O. Box 6199
Oceanside, CA 92058-6199

(760) 945-0613
art@caloxy.com
www.caloxy.com

TRADEMARK INFORMATION

SAS and SAS Quality Partner are registered trademarks of SAS Institute, Inc. in the USA and other countries.

® indicates USA registration.